

Programa en BASIC para el cálculo de grados días

J. L. GONZALEZ y J. M. HERNANDEZ

Se presenta un programa escrito en BASIC para microordenadores para el cálculo de grados días por los métodos de triangulación y función seno.

J. L. GONZALEZ y J. M. HERNANDEZ. I.N.I.A., Sección Estadística, Apdo. 8111. 28080 MADRID.

Palabras clave: Grados días, Basic.

INTRODUCCION

El ciclo biológico de muchos organismos depende estrechamente de la temperatura. Este hecho llevó al establecimiento del concepto de grados días, que ha sido utilizado con éxito en la predicción del desarrollo de diferentes organismos: Algodón (GUTIERREZ *et al.*, 1975); Alfalfa (GIESE *et al.*, 1975; Afidos (GILBERT y GUTIERREZ, 1973), etc.

En principio el método utilizado fue el número de grados-días en que la temperatura media diaria excede un determinado umbral inferior de temperatura. En este método, llamado de la temperatura media (WANG, 1960 y CROSS y ZOBEB, 1972), los grados días (GD) se estiman como sigue,

$$GD = (TMIN + TMAX)/2 - \text{Umbral inferior} \quad (1)$$

En 1956, LINDSEY y NEWMAN introducían un método donde los grados-días se calculaban con las áreas triangular o trapezoidal situadas entre la temperatura máxima y mínima diaria. Posteriormente, ARNOLD (1960) compara los métodos de triangulación, el de la media y un nuevo método trigonométrico, en el cual se ajusta una función seno a las temperaturas máximas y mínimas diarias. Encontrando

grandes errores en el método de la media y pequeños en los otros dos métodos. Este autor únicamente consideraba la existencia de un umbral inferior de desarrollo y no tenía en cuenta la situación donde altas temperaturas pueden retardar el desarrollo. Tanto SEVECHARIAN *et al.* (1977) por el método de triangulación, como ALLEN (1976) por el método de la función seno, eliminan la limitación anterior considerando la existencia de un umbral superior de desarrollo.

Para quienes estén interesados en profundizar en los métodos de cálculo de grados-días y su justificación, aconsejamos la lectura de WANG (1960); ALLEN (1976) y PRUESS (1983).

El objetivo de este trabajo es presentar un programa para microordenadores escrito en GWBASIC que permite realizar rápidamente el cálculo de los grados-días utilizando los dos métodos comúnmente utilizados, como son el método de triangulación y el método de la función seno.

METODO DE TRIANGULACION

SEVECHARIAN *et al.* (1977) modifican el método de triangulación de LINDSEY y NEWMAN (1956) considerando la existencia de un umbral superior. La figura 1




Fig. 1.—Rangos de temperaturas en relación con los umbrales superior e inferior. Las áreas sombreadas representan los grados-días.

presenta seis posibles rangos de temperaturas máximas y mínimas donde se considera la existencia de umbrales de temperatura. Las áreas sombreadas representan los grados-día computados para cada rango.

En el organigrama de la figura 2 presentamos el algoritmo utilizado para el cálculo de los grados-días correspondientes a los diferentes rangos de la figura 1. Para los rangos de temperaturas 1 y 5 (Fig. 1) el cálculo de grados-días son cero y el máximo respectivamente. Cuando la temperatura máxima está por encima del umbral superior, el área del triángulo superior D1 (rangos 4 y 6) se calcula utilizando el umbral superior H y la siguiente subrutina de triangulación:

$$D = (TMAX - T)^2 / (2 * (TMAX - TMIN))$$

Cuando se activan en el programa, el cálculo de los rangos 2 y la parte del rango 6 que está por debajo del umbral inferior, entonces se activa la subrutina de triangulación, calculándose posteriormente la diferencia entre el triángulo superior y el total del área que contiene los grados-días. El rango 3 se calcula con la ecuación 1.

METODO DE LA FUNCION SENO

La selección del rango en este método es idéntica a la utilizada en el método anterior. La única diferencia entre ambos métodos radica en la utilización de diferentes subrutinas, siendo para este método,

$$A = TMAX - M$$

$$B = T - M$$

$$Y = B/A$$

$$Z3 = ABS(1.0001 - Y^2)$$

$$Z2 = SQR(Z3)$$

$$Z1 = Y/Z2$$

$$Z = ATN(Z1)$$

$$D = (A * COS(Z) - b * ((M/2) - Z)) / P1$$

ABREVIATURAS UTILIZADAS

TMIN: Temperatura Mínima diaria.

TMAX: Temperatura Máxima diaria.

L: Umbral inferior de temperatura.

H: Umbral superior de temperatura.

U: Unidades de Grados días.

M: Temperatura Media diaria.

A: Amplitud de la función seno.

B, D, D1, T, Z: Variables temporales.

ATN: Arcotangente.

P1: 3,14159.

GD: Grados días.




Fig. 2.—Organigrama para el cálculo de grados-días por los métodos de triangulación y función seno.

ANEXO. Programa y salida del mismo

```

5 REM *****
10 REM *** PROGRAMA PARA CALCULAR GRADOS-DIAS***
20 PRINT "***** MENU *****"
30 PRINT "REALIZAR LOS CALCULOS DE GRADOS DIAS(1)"
40 PRINT "RETORNAR AL SISTEMA OPERATIVO (2)"
50 PRINT"*****":PRINT :PRINT
60 INPUT "INTRODUCIR EL NUMERO DE MENU DESEADO",MENU
70 ON MENU GOSUB 90,720
80 REM *****VALOR DE LOS PARAMETROS*****
90 P1=3.14159
100 CLS
110 INPUT "INTRODUCIR EL UMBRAL INFERIOR (L)",L
120 INPUT "INTRODUCIR EL UMBRAL SUPERIOR DE TEMPERATURA(H)",H
125 LPRINT "*****"
130 CLS:LPRINT "UMBRAL INFERIOR DE TEMPERATURA=";L"grados centigrados"
140 LPRINT "UMBRAL SUPERIOR DE TEMPERATURA=";H "grados centigrados"
150 REM *****INICIO DE LOS CALCULOS*****
160 IF H>L THEN GOTO 170 ELSE PRINT "ERROR, L>H":BEEP:GOTO 110
170 INPUT "INTRODUCIR TEMPERATURA MINIMA(TMIN)=" ,T1
175 INPUT "INTRODUCIR TEMPERATURA MAXIMA (TMAX)=" ,T2
180 U=0:IF T2>T1 THEN GOTO 190 ELSE PRINT "ERROR, MIN>=TMAX":BEEP:GOTO 160
190 M=(T1+T2)/2
200 S=T2-T1
210 CLS
220 LPRINT "*****"
230 LPRINT "TMIN=";T1,"TMAX=";T2,"TMEDIA=";M,"DIFERENCIA=";S
240 LPRINT"*****"
250 REM *****
260 REM *****METODO DE TRIANGULACION*****
270 REM *****
280 IF T2<L THEN GOTO 370 :REM *TEST PARA LOS RANGOS 1 Y 5
290 IF T1>H THEN GOTO 380
300 D1=0:IF T2<H THEN GOTO 330 ELSE 310:REM*SEPARACION RANGD 2 Y 3 DE 4 Y 6
310 T=H :GOSUB 420 :REM *AREA POR ENCIMA DEL TRIANGULO
320 D1=D
330 IF T1<L THEN GOTO 340 ELSE U=M-L-D1:GOTO 360:REM *RANGOS 3 Y 4
340 T=L:GOSUB 420 :REM *AREA TOTAL DEL TRIANGULO
350 U=D-D1:REM *RANGOS 2 Y 6
360 IF U>0 THEN GOTO 390
370 U=0: GOTO 420 :REM *RANGO 1
380 U=H-L :REM *RANGO 5
390 LPRINT "METODO DE TRIANGULACION ",INT(U*100+.5)/100:GOTO 460
400 REM
410 REM **SUBROUTINA PARA EL METODO DE TRIANGULACION**
420 D=(T2-T)^2/(2*(T2-T1)) : RETURN
430 REM *****
440 REM *****METODO DE LA FUNCION SENO*****
450 REM *****
460 IF T2<=L THEN GOTO 550:REM *TEST PARA RANGOS 1 Y 5
470 IF T1>H THEN GOTO 560
480 D1=0: IF T2<H THEN GOTO 510 *SEPARACION RANGOS 2 Y 3 DE 4 Y 6
490 T=H :GOSUB 590 :REM *AREA POR ENCIMA DEL TRIANGULO
500 D1=D
510 IF T1<L THEN GOTO 520 ELSE U=M-L-D1:GOTO 540:REM *RANGOS 3 Y 4
520 T=L :GOSUB 590 :REM *AREA TOTAL TRIANGULO
530 U=D-D1 :REM *RANGOS 2 Y 6
540 IF U>0 THEN GOTO 570
550 U=0 :GOTO 570 :REM *RANGO 1
560 U=H-L :REM *RANGO 5
570 LPRINT "METODO DE LA FUNCION SENO", INT(U*100+.5)/100:GOTO 650
580 REM **SUBROUTINA PARA EL METODO DE LA FUNCION SENO**
590 A=T2-M:B=T-M:Y=B/A

```

```

600 Z3=ABS(1.0001-Y^2)
610 Z2=SQR(Z3)
620 Z1=Y/Z2
630 Z=ATN(Z1)
640 D=(A*COS(Z)-B*((P1/2)-Z))/P1 : RETURN
650 LPRINT "*****"
660 PRINT:PRINT:PRINT
670 PRINT "PULSE CUALQUIER TECLA PARA VOLVER AL MENU"
680 IF INKEY$="" THEN GOTO 680
690 LPRINT:LPRINT:LPRINT
700 CLS
710 GOTO 20
720 SYSTEM
    
```

```

*****
UMBRAL INFERIOR DE TEMPERATURA= 8.5 grados centigrados
UMBRAL SUPERIOR DE TEMPERATURA= 23.7 grados centigrados
*****
TMIN= 9.8      TMAX= 31.1      TMEDIA= 20.45 DIFERENCIA= 21.3
*****
METODO DE TRIANGULACION      10.66
METODO DE LA FUNCION SENO    10.03
*****
    
```

```

*****
UMBRAL INFERIOR DE TEMPERATURA= 4.5 grados centigrados
UMBRAL SUPERIOR DE TEMPERATURA= 33 grados centigrados
*****
TMIN= 7      TMAX= 22      TMEDIA= 14.5 DIFERENCIA= 15
*****
METODO DE TRIANGULACION      10
METODO DE LA FUNCION SENO    10
*****
    
```

```

*****
UMBRAL INFERIOR DE TEMPERATURA= 10 grados centigrados
UMBRAL SUPERIOR DE TEMPERATURA= 38 grados centigrados
*****
TMIN= 6      TMAX= 23      TMEDIA= 14.5 DIFERENCIA= 17
*****
METODO DE TRIANGULACION      4.97
METODO DE LA FUNCION SENO    5.34
*****
    
```

```

*****
UMBRAL INFERIOR DE TEMPERATURA= 10 grados centigrados
UMBRAL SUPERIOR DE TEMPERATURA= 38 grados centigrados
*****
TMIN= 6      TMAX= 23      TMEDIA= 14.5 DIFERENCIA= 17
*****
METODO DE TRIANGULACION      4.97
METODO DE LA FUNCION SENO    5.34
*****
    
```

```

*****
UMBRAL INFERIOR DE TEMPERATURA= 6 grados centigrados
UMBRAL SUPERIOR DE TEMPERATURA= 18 grados centigrados
*****
TMIN= 7      TMAX= 17      TMEDIA= 12      DIFERENCIA= 10
*****
METODO DE TRIANGULACION      6
METODO DE LA FUNCION SENO    6
*****
    
```

ABSTRACT

GONZALEZ, J. L. y J. M. HERNANDEZ, 1990: Programa en BASIC para el cálculo de grados días. *Bol. San. Veg. Plagas*, **16** (1): 159-164.

A Basic computer program for calculating degree-day on microcomputer is presented the program calculates degree-day with the triangulation and sine wave methods.

Key words: Degree-day, Basic.

REFERENCIAS

- ALLEN, J. C., 1975: A modified sine wave method for calculating degree days. *Environmental Entomology*, **5**: 388-396.
- ARNOLD, C. Y., 1960: Maximum-minimum temperatures as a basis for computing heat units. *Journal of the American Society of Horticultural Science*, **76**: 682-692.
- CROSS, H. Z. y ZUBER, M. S., 1972: Prediction of flowering dates in maize based on different methods of estimating thermal units. *Agronomy Journal*, **64**: 351-355.
- GIESE, R. L.; PEART, R. M. y HUBER, R. Y., 1975: Pest management. *Science*, **187**: 1045-1052.
- GILBERT, N. y GUTIERREZ, A. P., 1973: A plant-aphidparasite relationship. *Journal of Animal Ecology*, **42**: 323-340.
- GUTIERREZ y col., 1975: An analysis of cotton production in California: A model for Acala cotton and the effects of the foliators on its yield. *Environmental Entomology*, **4**: 125-136.
- LINDSEY, A. A. y NEWMAN, J. E., 1956: Use of official weather data in spring time-temperature analysis of an Indiana phenological record. *Ecology*, **37**: 812-823.
- PRUESS, K. P., 1983: Day-degree methods for pest management, *Environmental Entomology*, **12**: 613-619.
- SEVECHARIAN, V. y col., 1977: Heat accumulation for timing lygus control measures in a saf-flower-cotton complex. *Journal of Economic Entomology*, **70**: 399-402.
- STARK, S. B. y ALINIAZEC, M. T., 1982: Evaluation of modifications to a basic thermal summation model for predicting the time of emergence of the aduct western cherry fruit fly. *Z. Angew. Entomology*, **94**: 401-407.
- WANG, J. Y., 1960: A critique of the heat unit approach to plant response studies. *Ecology*, **41**: 785-790.